BBSI Lab Session
June 25, 2003

The ODE's that are used are:

1) subdep.ode, example for substrate depletion

```
# substrate depletion
# dimensionless z=x+y
x'=nu*(z-x)*(eps^2+x^2)/(1+x^2)-x
z'=kappa-x
aux y=z-x
par kappa=0.2,eps=.05,nu=1
@ xp=x,yp=z,xlo=0.0,ylo=0.0,xhi=5,yhi=12,total=200
#  change dsmin to 1e-5 and dsmax to 0.1 in auto
# start with kappa=0.02 and intergate to get on a fixed point
#
done
```

2) actinh.ode, example for activator-inhibitor

```
# activator inhibitor
# gradually decrease b to 0.5 see the oscillation and then the fixed
point
x'=(eps^2+x^2)/((1+y)*(1+x^2))-a*x
y'=(b-y/(1+c*x^2))/tau
par a=.1,b=1,c=8,eps=0.1,tau=10
init x=.07811,y=1.04881
@ xp=x,yp=y,xlo=0,ylo=0,xhi=4,yhi=6
@ nmesh=100,total=1000,meth=qualrk,tol=1e-6,atol=1e-6
@ dt=.5
done
```

3) goldbeter.ode, model for circadian oscillations

```
# goldbeter model Proc Roy Soc B 261:319-324
#
m'=nus*1/(1+(pn/ki)^n)-num*m/(km+m)
p0'=ks*m-v1*p0/(p0+k1)+v2*p1/(p1+k2)
p1'=v1*p0/(p0+k1)+v4*p2/(k4+p2)-v2*p1/(k2+p1)-v3*p1/(k3+p1)
p2'=v3*p1/(k3+p1)-v4*p2/(k4+p2)-kk1*p2+kk2*pn-nud*p2/(kd+p2)
pn'=kk1*p2-kk2*pn
par nus=.76,num=.65,km=.5,ks=.38,nud=.95
par kk1=1.9,kk2=1.3,ki=1,kd=.2,n=4
par k1=2,k2=2,k3=2,k4=2
par v1=3.2,v2=1.58,v3=5,v4=2.5
# total protein
aux pt=p0+p1+p2+pn
#
init m=0.6,p0=.4,p1=.4,p2=.5,pn=.8
@ total=72
@ meth=qualrk,tol=1e-6
@ xp=pt,yp=m,xlo=0,ylo=0,xhi=72,yhi=6
done
```

4) tysoncirc.ode, 2-variable model for circadian oscillations

```
# tysons circadian model
# from  Biophys. J. 1999 77: 2411-2417
# this is the 2-variable model
q=2/(1+sqrt(1+8*keq*pt))
m'=num/(1+(.5*pt*(1-q)/a)^2)-km*m
pt'=nup*m-(kp1*pt*q+kp2*pt)/(jp+pt*(q+.5*r*(1-q)))-kp3*pt
par keq=200,num=1,km=.1,nup=.5,kp1=10,kp2=.03,kp3=0.1
par a=0.1,jp=.05,r=2
@ total=200
@ nmesh=100
@ xp=pt,yp=m,xlo=0,ylo=0,xhi=4,yhi=10
@ meth=qualrk,tol=1e-6
init m=1,pt=3.1
done
```

5) cycle3.ode, 4-variable cell cycle model with mass etc.

```
# cell cycle with mass and A
#

dx/dt=k1-(k2p+k2pp*y)*x
dy/dt=-k4*m*x*y/(j4+y)+(k3p+k3pp*a)*(1-y)/(j3+1-y)
da/dt=k5p+k5pp*(m*x)^n/(j5^n+(m*x)^n)-k6*a
dm/dt=mu*m*(1-m/mstar)
par k1=.04,k2p=.04,k2pp=1
par k3p=1,k3pp=10,k4=35,j3=.04,j4=.04
par k6=.1,k5pp=.2,k5p=.005,j5=.3,n=4
par mu=.01,mstar=10
par xthr=0.1
# when x drops to xthr, cell divides and mass (m) becomes half
global -1 x-xthr {m=.5*m}
init x=.1,y=.9,a=.2,m=.6
@ total=150,dt=.05,meth=qualrk,tol=1e-6,atol=1e-6
@ nplot=2,xhi=150,yp=x,yp2=m
@ xp2=t,xp3=t,xp4=t,ylo=0,yhi=1.5
done
```

The results of these ODE's:

1) subdep.ode, example for substrate depletion

kappa=0.2
i) Commands: Initialconds, Go, Nullcline, New



Figure 1. Note that fixed point is unstable (using initialconds, mice)

ii) Commands: Param, kappa=0.02, ok, Nullcline, New (x=0.02 is the invisible nullcline ☺)



Figure 2. Note that the new fixed point is stable (using initialconds, mice)

iii) Commands: Param, kappa=1, ok, Nullcline, New



Figure 3. Note that the new fixed point is stable (using initialconds, mice)

2) actinh.ode, example for activator-inhibitor

i) b=1

Commands: Nullcline, New, Initialconds, Mice



Figure 4. The fixed point is unstable

ii) Commands: Param, b=0.8, ok, Nullcline, New, Initialconds, Mice



Figure 5. The fixed point is unstable which leads to oscillations

iii) Commands: Param, b=0.5, ok, Nullcline, New, Initialconds, Mice



Figure 6. The fixed point is stable

3) goldbeter.ode, model for circadian oscillations

i) Commands: Initialconds, go



Figure 7. M (mRNA) versus PT (total PER protein)

ii) Commands: erase, Xi vs t, M



Figure 8. M versus time (note that it is periodic for 24 hours)

iii) Commands: erase, Xi vs t, PT



Figure 9. PT versus time (note that it is periodic for 24 hours)

4) tysoncirc.ode, 2-variable model for circadian oscillations

i) Commands: Initialconds, go



Figure 10. M (mRNA) versus PT (total PER protein)

ii) Commands: erase, Xi vs t, M



Figure 11. M versus time (note that it is periodic for 24 hours)

iii) Commands: erase, Xi vs t, PT



Figure 12. PT versus time (note that it is periodic for 24 hours)

5) cycle3.ode, 4-variable cell cycle model with mass etc.

Commands: Initialconds, go



Figure 13. Black curve is x (cyclin CDK) vs. time and red curve is m (mass) vs. time. Mass becomes half (cell divides) when x drops to a certain value.